UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

A Program in Hewlett-Packard BASIC for plotting rare-earth element diagrams, using HP-Series 80 computers

by

J. S. Stuckless[1/] and E. H. Christiansen[2/]

[1/] U.S. Geological Survey, Denver Federal Center, MS 963, Denver, CO 80225

[2/] Department of Geology, University of Iowa, Iowa City, Iowa 52242

Open-File Report 84-123

1984

A program in Hewlett-Packard BASIC for plotting rare-earth element diagrams, using HP-Series 80 computers

INTRODUCTION

The rare-earth elements (REE) are the fifteen elements in the range of atomic numbers 57 through 71. (One of these REE, promethium, is radioactive with a short half-life, and it does not occur in terrestrial rocks.) All of the REE have very similar chemical properties in the +3 valance state, and thus fractionation between adjacent rare earth is generally very small. Europium is unique in that a valance of +2 is common, and therefore, marked fractionation between europium and its adjacent rare earths does occur. Because of these properties, REE are used extensively in petrologic studies to assist in modeling magma derivation and evolution (for example, Hanson, 1978).

The cosmic abundances of REE generally decrease with increasing atomic number in a saw-tooth pattern such that the even-numbered REE are more abundant than adjacent odd numbered REE (Ahrens, 1965). Haskin and Gehl (1962) noted that REE abundances in terrestrial rocks divided by REE abundances in chondrite meteorites produced smooth patterns if the resulting quotients are plotted as an arithmetic function of atomic number. Currently, most geologic reports display REE data as normalized to a common base such as abundances in chondrite meteorites (Evensen and others, 1978), or a least- evolved member of a rock suite (Sun and Hanson, 1976). Data for minerals are typically normalized to concentrations of the host rock (Hanson, 1978).

The program "REEPLOT" described here is written in HP-BASIC and produces semilogarithmic plots of REE data which may be displayed on a cathode-ray tube (CRT), an external plotter, or a dot-matrix printer. Data may be entered from the keyboard, from files stored by the electronic worksheet program "VISICALC", or from a combination of the two methods. Plotted data can be represented by any alpha-numeric symbol, open or solid polygons or error bars. Solid lines are used to join data points, but if poorly known or estimated values are entered as negative numbers, dashed lines are used through points with the corresponding positive values. The program can also be used to estimate the concentration of an unreported REE. Data plots can be normalized to chondrite values of Evensen and others (1978), a set of values stored with the data, or data entered from the keyboard by the user.

EQUIPMENT REQUIREMENTS

The program is written for use with Hewlett-Packard series 80 computers and may require modifications for use with other computers that operate in BASIC. Disc storage requires about 55 records of 256-byte length. Operation of the program require approximately 40,000 bytes of memory. The program accepts up to 75 sets of data.

Plotting may be done on a cathode-ray tube (CRT), a hard-copy plotter (such as the HP-7470A), or a graphic printer (such as the HP-82905A). This last option requires either installation of a plotter ROM, or prior loading into memory of the binary program "GDUMP" (which is available on the Hewlett-Packard Demonstration Disc). Some printers must be reset after each graphic dump. In such cases, a command of "RESET [interface select code]" must be added at the

end of the "DUMP GRAPHICS" subroutine and before the return from the subroutine. For example, rewrite line 2430 (Appendix I) to read: "DUMP GRAPHICS 0,0,-1,I @ RESET 9 @ GOTO 1730".

PROGRAM OPERATION

The program executes interactively through a series of queries which are largely self-explanatory. All queries are answered by typing a response and pressing the "END LINE" key. Multiple answers to a single query should be separated by commas. All alpha answers should use upper case letters.

The first query displayed is "PLEASE SPECIFY PRINTER ADDRESS". This query is also displayed after each request for printed output. The input of an impossible printer address will cause the query to be redisplayed. The input of a possible address will be accepted even if the requested device is not turned on.

The second query asks for the source of data to be plotted: "VISICALC DATA FILE (V) OR KEYBOARD DATA ENTRY (K)". A response of "K" causes a series of queries as to which REE will be input: "DO YOU HAVE DATA FOR _____ (YES OR NO)?" where the blank is filled sequentially with the chemical symbol for each of the possible REE. Any response other than "Y" will cause the program to assume that there are no data for the REE displayed in the current query. After the query for lutetium (Lu) has been answered, the program displays "SAMPLE NUMBER? ENTER 'END' WHEN THROUGH". The sample number can be any alpha-numeric combination up to 18 characters in length; however, the print out provides space for only the first 7 characters. The image statements in lines 850 and 860 (Appendix I) must be modified if printouts of longer sample names are desired. Entry of any sample name other than "END" causes the query: "SAMPLE (identification), (REE name) =?" for each REE specified in the first series of queries. An answer of "END" moves the program to subsequent program segments.

If the data input query is answered by "V", the program displays restrictions for the format of the "VISICALC" file: "DATA AND SAMPLE NAMES MUST BE WITHIN THE FIRST 15 COLUMNS AND DATA COLUMNS MUST BE HEADED BY A CHEMICAL ELEMENT SYMBOL. e.g. La". An example of a usable "VISICALC" file is shown on figure 1. The use of standard chemical symbols (upper case first letter and lower case second letter) as column headings is essential because the program uses the symbols to identify values for specific REE. Other information in addition to the chemical symbols such as a data-set title or units for the data can appear above the data (Fig. 1). This information will be ignored during the data reading.

If a "VISICALC" input has been requested, the display will also show "ENSURE THAT FILE IS IN CURRENT STORAGE MEDIUM BEFORE ENTERING FILE NAME. ENTER FILE NAME, OR PRESS [END LINE] FOR CATALOG." If the requested file is not found, the file name is displayed followed by: "FILE DOES NOT EXIST! DO YOU NEED TO CHANGE MASS STORAGE?: (YES OR NO)." Any alpha-numeric answer other than "Y" will cause the query for a file name to be displayed. Pressing "END LINE" without any input will restart the program. An answer of "Y" will cause a display of: "MASS STORAGE IS '_____'?", and input of a proper mass storage address will cause a new attempt to locate the previously requested file. Once the requested data file is located, the program assumes that any sample

identification information will be to the left of the data. The first data row and column are obtained from the query: "WHAT IS THE FIRST ROW AND FIRST COLUMN WITH DATA? (e.g. 4, 2)". The answer "4, 2" would be appropriate for the data file shown on figure 1. An error will result during the data read operation if any non-numeric data are encountered in or below the specified row or in or to the right of the specified column. Null data (blanks) are accepted and converted to values of zero.

The order in which the elements appear in the data file is not important because the program will reorder the data to the order and spacing needed for plotting. After the data columns have been placed in the correct order for plotting, the program displays: "DO YOU WANT THE DATA PRINTED? (YES OR NO)". An example of the printed data is shown on figure 2. The row number in this printout is used to identify samples in subsequent subroutine, and therefore it is generally advisable to make a hard-copy print of the data. After the data have been printed or if any answer to the query other than "Y" is input, the program displays: "ARE DATA OK AND READY TO PLOT? (YES OR NO)". Any answer other than "Y" will activate a subroutine to edit the data. Values to be changed are identified by a row number for the samples (as indicated in the printout of the data, fig. 2) and by the chemical symbol of the element (2 upper-case letters or an upper and lower cased letter can be accepted, for example, "LA" or "La").

The final data set can be normalized through use of any of three options. The program displays "CHONDRITE (C) OR OTHER (O) NORMALIZED PLOTS?". An answer of "C" will normalize the data to the chondrite values of Evensen and others (1978). Any other answer will cause a display of "INPUT ROW NUMBER FOR NORMALIZING VALUES OR K FOR KEYBOARD ENTRIES". If keyboard entry is requested, values will be requested for only the REE identified in the current data set. After the data are normalized, a query provides an option to print the results (fig. 3).

A plotting base is created for the normalized data by first displaying: "ENTER 1 FOR CRT PLOT OR PLOTTER ADDRESS". A choice of pen numbers is then provided. This query is followed by: "MIN AND MAX VALUES? ENTER EQUAL NUMBERS IF UNKNOWN". The MIN and MAX values refer to the minimum and maximum values of the normalized data (fig. 3). If these limiting values are not known, the program will search for even decade limits. If even decade axis tics cannot be found with user-supplied limits, the program will ask for even decade limits (for example, 10 and 100).

The outline of the plotting base is drawn such that the Y-axis (the logarithms of the normalized vlaues) is in the long dimension of the plotter. The Y-axis is marked logarithmically and labelled with anti-logarithmic values at decades of 5 and 10. The X-axis is marked at 15 even spaces. The program then displays "LABEL ONLY REPORTED REE? (YES OR NO)". Any alpha-numeric answer other than "Y" will cause the X-axis to be labeled with all naturally occurring REE (fig. 4). An answer of "Y" will result in chemical symbols labelled for only REE in the data set in use (fig. 5 and 6). X and Y axes are titled "REE ATOMIC NUMBER" and "SAMPLE/CHONDRITE", respectively, unless data were normalized to other than the stored chondrite values, in which case, a title for the Y-axis is requested. An option is provided to place a title at the top of the plot (figs. 6 and 7). If commas are to be used in the title (fig. 6), the input must be enclosed in quotation marks or an error of too many inputs will result. If no title is desired, enter a space.

RARE-EARTH TEST SET NOV. 24, 1983

SAMPLE NAME	La (ppm)	Ce (ppm)	Nd (ppm)	Sm (ppm)	Eu (ppm)	Gd (ppm)	Tb (ppm)	Dy (ppm)	Tm (ppm)	Yb (ppm)	Lu (ppm)
77735	35.08	76.4	37.33	8.52	.39	7.64	1.62	12.07	1.69	11	1.78
77738	82.58	170.4	76.7	17	1.84	15.9	2.93	21.55	2.85	19.4	3.08
77662	75.55	143.5	58.83	11.96	.98	10.88	2.26	16.7	2.3	15.7	2.44
77666	52.8	99.85	45.7	9.65	.399	8.96	1.67	11.97	1.58	10.9	1.69
77735	35.08	76.4	37.33	8.52	-.39	7.64	1.62	12.07	1.69	-11	1.78

Figure 1.--Printout of a "VISICALC" file used as input for the program "REEPLOT".

Lu	Yb	Tm	Dy	Tb	Gd	Eu	Sm	Nd	Ce	La	ROW	SAMPLE
1.78	11.00	1.69	12.07	1.62	7.64	.39	8.52	37.33	76.40	35.08	1	77735
3.08	19.40	2.85	21.55	2.93	15.90	1.84	17.00	76.70	170.40	82.58	2	77738
2.44	15.70	2.30	16.70	2.26	10.88	.98	11.96	58.83	143.50	75.55	3	77662
1.69	10.90	1.58	11.97	1.67	8.96	.40	9.65	45.70	99.85	52.80	4	77666
1.78	-11.00	1.69	12.07	1.62	7.64	-.39	8.52	37.33	76.40	35.08	5	77735

Figure 2.--Printout of data produced by the program REEPLOT.

Lu	Yb	Tm	Dy	Tb	Gd	Eu	Sm	Nd	Ce	La	ROW	SAMPLE
74.48	66.63	65.99	47.50	43.26	37.40	6.72	55.32	78.79	119.77	143.42	1	77735
128.87	117.50	111.28	84.81	78.24	77.83	31.71	110.39	161.88	267.13	337.61	2	77738
102.09	95.09	89.81	65.72	60.35	53.26	16.89	77.66	124.17	224.96	308.87	3	77662
70.71	66.02	61.69	47.11	44.59	43.86	6.88	62.66	96.45	156.53	215.86	4	77666
74.48	.02	65.99	47.50	43.26	37.40	.19	55.32	78.79	119.77	143.42	5	77735

Figure 3.--Printout of chondrite-normalized data produced by the program REEPLOT.



Figure 4.--REE diagram for rows 2 and 4 (fig. 3). Row 2 was plotted without a plot symbol request. Row 4 was plotted using the error bar option. Errors used were: La=3%, Ce=2%, Nd=3%, Sm=4%, Eu=5%, Gd=8%, Tb=6%, Dy=4%, Tm=6%, Yb=4%, and Lu=6%.



Figure 5.--REE diagram for rows 1 and 3 (fig. 3). Row 1 was plotted using alpha-numeric symbol "X". Row 3 was plotted using the polygon option. Only REE in the input data set are labelled.



Figure 6.--REE diagram for rows 2, 3, and 5 (fig. 3). Row 2 was plotted using alpha-numeric symbol "0". Row 3 was plotted using the polygon option. Row 5 was plotted without use of a plotting symbol; dashed lines extend to and from data points entered as negative values. Only reported REE are labelled and the title option was used.

If the CRT is used for plotting, the plot base will be stored to the currently active storage medium under the name "OUTLINE". Storage of the plot base requires approximately 50 records of 256 byte length and if insufficient room exists, an error will result. An option to replace the current graphics display with the stored plot base is offered after each request for a plot of a data set as long as the CRT is identified as the current plot device.

Data to be plotted on the base are identified by the row numbers as given in the data printout (fig. 2) and must be entered one set at a time. The query "ROW TO BE PLOTTED?" must have a response of a single numeric input. This query is followed by a choice of pen number and then "DO YOU WANT A PLOT SYMBOL FOR DATA POINTS? (YES OR NO)". Any answer other than "Y" will produce a line plot with no data symbols as shown on figure 4 for the data set with the higher REE contents. Solid lines are used to connect data points (whether points are actually shown or not) unless data are entered as negative numbers (for example, sample 77735, row 5, fig. 2). Negative numbers might be used for poorly known or estimated values. The absolute value is used to locate the position of the data point, but the line to and from that data point will be dashed (fig. 6 and 7). If data points are shown by symbols, no symbol will be plotted for negative input (fig. 7).

If a plot symbol is requested, the program will display "SYMBOL CAN BE ANY ALPHA-NUMERIC KEY OR 'P' FOR A POLYGON OR 'E' FOR ERROR BARS." If "X" or "O" are request, the aspect ratio will be adjusted such that height and width will be equal (fig. 5 and 6). Note that "O" is barely distinguishable from a square (fig. 6).

A response of "E" to the plot-symbol question will produce a query for each element for which there are data: "WHAT IS THE 2 SIGMA ERROR (%) FOR _____" where the blank is filled sequentially with the chemical symbol for the REE. An example of an error-bar plot is shown on figure 4. Errors used in the example range from 2 to 8 percent.

A response of "P" to the plot-symbol question produces the query "# OF SIDES, OPEN OR SOLID (O OR S), SIZE (mm), AND ROTATION (0 TO 360)". An example of polygon plot-symbols and the necessary input for each is presented on figure 8. Polygon plot-symbols are used for REE diagrams on figures 5 through 7. A request for a polygon with less than 3 sides will cause the query for input to be redisplayed. The size parameter is interpreted in millimeters by external plotters and as relative size for CRT plots. As a reference, the numbers on the Y-axis are size 3 and the X-axis title is size 4.

After the data have been plotted for a given sample (or a continue command has been given if the CRT has been used for plotting), seven options are displayed:

"DO YOU WANT TO CHANGE PLOTTER OR REDRAW BASE? (Y)"
"DO YOU WANT TO FIND A VALUE FOR A REE ON THE CURRENT PLOT? (V)" (this can only be done on external plotter)"
"DO YOU WANT A DOT-MATRIX PRINT OF DIAGRAM? (P)"
"DO YOU WANT TO ADD DATA FROM THE KEYBOARD? (K)"
"DO YOU WANT TO EDIT EXISTING DATA? (E)"
"DO YOU WANT TO CHANGE NORMALIZATION FACTOR (N)"
"DO YOU WANT TO CONTINUE WITHOUT CHANGES? (C)"



Figure 7.--REE diagram for rows 2 and 5 (fig. 3). Both data sets were plotted using the polygon option; however, negative values for Eu and Yb in sample 5 cause suppression of the plotting symbol and use of dashed line through the positive equivalent of these values. Optional title is also used.



Figure 8. Plotting symbols generated by the polygon command and the input used to draw each.

Options are chosen by responding with the letter in parentheses after the desired option. The default option is to continue without changes and therefore any response other than "Y", "V", "P", "K", "E", or "N" will produce a query for the next set to be plotted. Responses of "Y", "K", or "E" access subroutines which have been discussed previously.

The option to change the normalization factor returns the program to the query "CHONDRITE (C) or OTHER (O) NORMALIZED PLOTS?" discussed previously. After new ratios and logarithms have been calculated, an option to print the data is displayed This option is followed by the query "ARE DATA OK AND READY TO PLOT? (YES OR NO)". Any answer other than "Y" will send the program to the subroutine to edit data. An answer of "Y" will send the program to the subroutine to create a new plotting base.

The option to find a value for a REE can only be used with plots drawn by an external plotter and that plotter must have a digitizing capability. The process starts with a query to identify the printing devise for output of calculated results. This is followed by the query: "FOR WHICH REE DO YOU WANT A VALUE? ENTER 'END' WHEN FINISHED". The response can be an upper and lower case letter "La" or two upper case letters "LA", but at least one sample of the data set must have a value for the element requested. An improper response will cause the query to be redisplayed. A response of "END" will regenerate the display of the seven options discussed above. After a REE identifier has been entered, the pen in current use (or digitizing site if one is available) will be moved to the point on the X-axis that corresponds to the requested REE, and the program will display: "MOVE PEN TO DESIRED VALUE OVER PLOT" which should be done by use of the Y-axis control only, and "PRESS 'ENTER' ON PLOTTER TO CONTINUE".

If the option for a dot-matrix print of the CRT display is chosen, the program will display: "INPUT PRINTER TYPE: 0=82905B, -1=82905A, 1=MOST OTHERS". Operation of the dot matrix print option requires either a plotter ROM or prior loading of the binary program "GDUMP" which is available on the Hewlett-Packard demonstration disc. Some printers may need to be reset after the graphics print as described in the Equipment Requirement section. If an improper printer type is entered, the attempt to copy the CRT display will produce nonsense. After the CRT graphics have been copied, the CRT display returns to the seven options discussed above.

REE PLOT is written to recover from a few common errors such as file name or mass storage errors. Some other errors, such as too many inputs, can be recovered by the user because the program will display the error number and the line on which the error occurred, and the program will pause. For example, if the title for figure 6 is entered without enclosing quotation marks, the program will display: "ERROR NUMBER 45 ON LINE 1360" and "PROGRAM IS PAUSED". Entering "CONT 1360" will cause the request for a plot title to be redisplayed and after an acceptable plot title has been entered, the program will continue operation. A complete program listing is presented in Appendix I.

REFERENCES

Ahrens, L. N., 1965, Distribution of the elements in our planet: McGraw-Hill Book Co., 110 p.

Evensen, N. M., Hamilton, P. J., and O'Nions, R. K., 1978, Rare earth abundances in chondritic meteorites: Geochimica et Cosmochimica Acta, v. 42, p. 1199-1212.

Hanson, G. N., 1978, The application of trace elements to the petrogenesis of igneous rocks of granitic composition: Earth and Planetary Science Letters, v. 38, p. 26-43.

Haskin, L. H. and Gehl, M. A., 1962, The rare-earth distribution in sediments: Journal of Geophysical Research, v. 67, p. 2537-2541.

Sun, S. S. and Hanson, G. N., 1976, Rare earth element evidence for differentiation of McMurdo volcanics, Ross Island, Anarctica: Contributions to Mineral Petrology, v. 54, p. 139-155.

APPENDIX I

```
10 REM RARE-EARTH ELEMENT PLOTTING PROGRAM BY J. S. STUCKLESS MAY 22,1983
20 REM CHONDRITE ABUNDANCES FROM EVENSEN ET. AL, 1978, GEOCHIM. COSMOCHIM. ACTA,
V. 42, P. 1203
30 DIM B$[151],S(75,15),N$(75),L$(15),A$(15),X(15),X$[10],REE$[2],L(15),K(15),T(
75,15)
40 INTEGER RMIN,RMAX,CMIN,CMAX
50 F1,Q9,RMAX,CMAX,RMIN,CMIN=0 @ FOR I=1 TO 15 @ L$(I)="" @ NEXT I
60 I$=" ("&CHR$ (217)&"ES OR "&CHR$ (206)&"O)" @ CLEAR @ GOSUB 330
70 RESTORE 1370 @ FOR I=15 TO 1 STEP -1 @ READ A$(I)@ NEXT I
80 DISP "VISICALC DATA FILE (V) OR KEYBOARD DATA ENTRY (K) ?" @ INPUT A$
90 IF A$="K" THEN 2230
100 CLEAR @ IF A$#"V" THEN 80 ! DISPLAY PROGRAM RESTRICTIONS ******************
110 DISP "DATA AND SAMPLE NAMES MUST BE WITHIN THE FIRST 15 COLUMNS AND" @ DISP
120 DISP "DATA COLUMNS MUST BE HEADED BY A CHEMICAL ELEMENT SYMBOL. e.g. La."
130 GOSUB 280 @ INPUT A$
140 IF NOT LEN (A$) THEN GOSUB 300 @ CLEAR @ GOTO 130
150 DISP "WHAT IS THE FIRST ROW AND FIRST COLUMN WITH DATA ? (e.g. 4,2)"
160 INPUT RMIN,CMIN@ R=RMIN-1 @ ON ERROR GOTO 250
170 ASSIGN# 1 TO  A$ @ CLEAR
180 DISP "File name: "&A$&" is being loaded" @ ON ERROR GOTO 190 @ GOTO 390
190 OFF ERROR @ IF ERRN =71 OR ERRN =72 THEN 230
200 IF ERRN =89 THEN DISP "INVALID DATA ENCOUNTERED IN ROW";I;@ GOTO 80
210 DISP "ERROR NUMBER ";ERRN ;" ON LINE ";ERRL
220 BEEP @ DISP "PROGRAM IS PAUSED " @ PAUSE
230 FOR K=1 TO RMAX-RMIN+1 @ N$(K)=N$(K+R) @ NEXT K
240 ASSIGN# 1 TO  * @ GOTO 560
250 OFF ERROR @ BEEP @ DISP A$;" FILE DOES NOT EXIST! DO YOU NEED TO CHANGE MASS
 STORAGE?:"&I$ @ INPUT A$@ IF NOT LEN (A$) THEN CLEAR @ GOTO 80
260 IF UPC$ (A$[1,1])#"Y" THEN CLEAR @ GOTO 130
270 DISP "MASS STORAGE IS '____'";@ INPUT M$@ MASS STORAGE IS M$ @ GOTO 170
280 DISP USING "3/" @ DISP "ENSURE THAT FILE IS ON THE CURRENT STORAGE MEDIUM BE
FORE ENTERING FILE NAME" @ DISP
290 DISP "ENTER FILE NAME, OR PRESS [END LINE] FOR CATALOG" @ RETURN
300 CRT IS 2 @ DISP @ CAT @ DISP USING "3/" @ CRT IS 1
310 DISP @ DISP "* PRESS [CONT] KEY WHEN READY *" @ PAUSE
320 RETURN
330 CLEAR @ GOTO 350 ! SELECT PRINTER SUBROUTINE ********************************
340 CLEAR @ DISP @ DISP "INCORRECT PRINTER SPECIFICATION" @ DISP
350 DISP @ DISP "PLEASE SPECIFY THE PRINTER ADDRESS." @ DISP
360 DISP "1 = the CRT, OTHER = NUMBER OF THE EXTERNAL PRINTER" @ DISP
370 ON ERROR GOTO 340 @ DISP @ INPUT Q7@ PRINTER IS Q7 @ PRINT @ OFF ERROR
380 RETURN
390 READ# 1 ; B$@ IF POS (B$[1],"/") THEN 230
400 IF NOT POS (B$[1],">") THEN 390 ! SUBROUTINE LOAD 'VISICALC' ************
410 I=VAL (B$[3]) @ IF I>RMAX THEN RMAX=I
420 J=NUM (B$[2])-64 @ IF J>CMAX THEN CMAX=J
430 Z=POS (B$,":")+1 @ IF NUM (B$[Z])=34 THEN Z=Z+1
440 IF POS (B$[Z]," ") THEN Z=Z+1 ELSE 460
450 GOTO 440
460 IF J<CMIN THEN PU=LEN (B$[Z]) ELSE 480
470 N$(I)=B$[Z,Z+PU-2] @ S(I,J)=0 @ GOTO 390
480 IF I<RMIN THEN 500
490 S(I-R,J)=VAL (B$[Z]) @ GOTO 390
500 FOR I=1 TO 15 @ IF POS (B$,A$(I)) THEN 520
510 NEXT I @ GOTO 390
520 L$(J)=B$[Z,Z+1] @ GOTO 390 ! REORDER COLUMNS TO PLOTTING ORDER ************
```

```
530 OFF ERROR @ IF ERRN #7 THEN GOTO 190 ELSE S(I,X9)=0 @ RETURN
540 FOR I=1 TO RMAX @ ON ERROR GOSUB 530 @ T(I,J)=S(I,X9) @ NEXT I @ T(I,J)=1 @
RETURN
550 T(I,J)=S(I,X9) @ NEXT I @ T(I,J)=1 @ RETURN
560 CLEAR @ DISP "ONE MOMENT PLEASE WHILE DATA ARE REORGANIZED"
570 RMAX=RMAX-RMIN+1 @ X=RMAX+1 @ FOR J=1 TO 15 @ T(X,J)=0 @ NEXT J
580 FOR X9=1 TO 15 @ FOR J=1 TO 15 @ IF J=11 THEN 600
590 IF L$(X9)=A$(J) THEN GOSUB 540 @ GOTO 610
600 NEXT J @ ON ERROR GOTO 190
610 NEXT X9 @ FOR J=1 TO 15 @ IF T(X,J)=1 THEN L$(J)=A$(J) ELSE L$(J)=""
620 IF L$(J)="" THEN GOSUB 650 @ GOTO 640
630 FOR I=1 TO RMAX @ S(I,J)=T(I,J) @ NEXT I @ T(I,J)=0
640 NEXT J @ RMIN=1 @ GOTO 660
650 FOR I=1 TO RMAX @ S(I,J)=0 @ NEXT I @ RETURN
660 GOSUB 680 @ GOSUB 900
670 IF Q9=1 THEN RETURN ELSE GOSUB 680 @ GOTO 1080
680 CLEAR @ DISP "DO YOU WANT DATA PRINTED ?"&I$ @ INPUT A$@ IF A$#"Y" THEN 780
690 GOSUB 330 @ X9=11 ! ************** PRINT DATA ******************************
700 FOR J=1 TO 15 @ IF L$(J)="" THEN 720
710 A$(X9)=L$(J) @ X9=X9-1
720 NEXT J
730 PRINT USING 870 ; A$(11);A$(10);A$(9);A$(8);A$(7);A$(6);A$(5);A$(4);A$(3);A$
(2);A$(1)
740 FOR I=1 TO RMAX @ X9=0 @ FOR J=1 TO 15 @ IF L$(J)="" THEN 760 ELSE X9=X9+1
750 IF F1=0 THEN X(X9)=S(I,J) ELSE X(X9)=10^T(I,J)
760 NEXT J
770 PRINT USING 880 ; X(1),X(2),X(3),X(4),X(5),X(6),X(7),X(8),X(9),X(10),X(11),I
,N$(I) @ X9=1 @ NEXT I
780 DISP @ DISP "ARE DATA OK AND READY TO PLOT ?"&I$ @ INPUT A$@ IF A$="Y" THEN
RETURN
790 CLEAR ! ********************** EDIT DATA **********************************
800 DISP "EDIT VALUES IN ROW NUMBER ___?" @ INPUT I
810 IF I<RMIN OR I>RMAX OR I#INT (I) THEN DISP "INVALID ROW #" @ GOTO 800
820 DISP "EDIT REE 'SYMBOL' ?" @ INPUT A$
830 FOR J=1 TO 15 @ IF UPC$ (A$)=UPC$ (L$(J)) THEN 850
840 NEXT J @ IF J=16 THEN DISP "REE NOT FOUND IN CURRENT DATA SET. TRY AGAIN." @
 GOTO 780
850 DISP "ROW";I;"SAMPLE ";N$(I);" ";L$(J);"=" @ INPUT S(I,J)
860 IF Q9=2 THEN RETURN ELSE GOTO 780
870 IMAGE "   ",10(AAAAAA),AAAA," ROW ","SAMPLE"
880 IMAGE 11(DDD.DD),X,DDD,XX,7A
890 DATA .0239,.1651,.02561,.166,.0567,.2541,.03745,.2043,.05802,.154,1,.4738,.0
9637,.6379,.2446
900 CLEAR @ DISP "CHONDRITE (C) OR OTHER (O) NORMALIZED PLOTS ?" @ INPUT A$
910 IF A$#"C" THEN F1=2 @ GOTO 930 ELSE F1=1 ! NORMALIZE DATA ******************
920 RESTORE 890 @ FOR I=1 TO 15 @ READ L(I)@ NEXT I @ GOTO 1000
930 DISP "INPUT ROW NUMBER FOR NORMALIZING VALUES OR K FOR KEYBOARD ENTRIES."
940 INPUT A$@ IF A$="K" THEN 970
950 IF VAL (A$) THEN I=VAL (A$) ELSE 930
960 FOR J=1 TO 15 @ L(J)=S(I,J) @ NEXT J @ GOTO 1000
970 FOR J=1 TO 15 @ IF L$(J)="" THEN 990
980 DISP L$(J);"NORMALIZING VALUE =" @ INPUT L(J)
990 NEXT J
1000 CLEAR @ DISP "TAKING RATIOS AND LOGS, A FEW MOMENTS PLEASE" @ DISP
1010 FOR J=1 TO 15 @ IF L$(J)="" THEN 1070
1020 FOR I=RMIN TO RMAX @ IF A$="O" OR A$="K" THEN 1040
1030 IF S(I,J)<0 THEN T(I,J)=-LGT (ABS (S(I,J))/L(J)) @ GOTO 1060
1040 IF S(I,J)=0 THEN T(I,J)=S(I,J) @ GOTO 1060
1050 T(I,J)=LGT (S(I,J)/L(J))
1060 NEXT I
```

```
1070 NEXT J @ RETURN ! ****** CREATE PLOTTING BASE ****************************
1080 CLEAR @ DISP "ENTER 1 FOR CRT PLOT OR PLOTTER ADDRESS";@ INPUT Q8
1090 PLOTTER IS Q8 @ IF Q8=1 THEN LIMIT 0,125.6,0,75.2 @ GCLEAR @ GOTO 1110
1100 LIMIT 0,257,0,191
1110 LOCATE 20,120,5,85 @ T=.2 @ U=0 @ CLEAR
1120 DISP "PEN NUMBER" @ INPUT A@ PEN A
1130 DISP "MIN AND MAX VALUES ? ENTER EQUAL NUMBERS IF UNKNOWN." @ INPUT A,B
1140 IF A=B THEN GOSUB 1430
1150 A1=A @ GOTO 1580
1160 SCALE LGT (A),LGT (B),0.16 @ S1=S @ A3=5*S
1170 IF 10*A>B THEN U1=B @ GOTO 1190
1180 U1=10*S @ GRAPH
1190 FOR X=A TO U1 STEP S @ PLOT LGT (X),U,-1 @ PLOT LGT (X),U+T
1200 PLOT LGT (X),U,1 @ IF T>0 THEN 1240
1210 IF X/A3#X\A3 THEN 1240
1220 CSIZE 3 @ LORG 8 @ DEG @ LDIR 270 @ X$=VAL$ (X) @ X$=X$&" "
1230 LABEL USING "K" ; X$ @ PLOT LGT (X),U,1
1240 NEXT X
1250 S=S*10
1260 A=S @ IF T<0 THEN A3=A3*10
1270 IF A<B THEN 1170
1280 IF T<0 THEN 1310
1290 MOVE LGT (A1),0 @ N=(LGT (B)-LGT (A1))/191
1300 YAXIS LGT (A1),1,0,16 @ T=-T @ U=16 @ A=A1 @ S=S1 @ GOTO 1170
1310 YAXIS LGT (B),1,0,16 @ A4=LGT (A1)-6*N
1320 DISP "LABEL ONLY REPORTED REE ?"&I$ @ INPUT A$
1330 LORG 6 @ DEG @ LDIR 270 @ FOR I=15 TO 1 STEP -1 @ MOVE A4,I
1340 IF A$="Y" THEN 1350 ELSE READ REE$@ LABEL REE$ @ GOTO 1360
1350 LABEL L$(I)
1360 NEXT I @ CSIZE 4 @ N1=LGT (B)+7*N
1370 DATA La,Ce,Pr,Nd,,Sm,Eu,Gd,Tb,Dy,Ho,Er,Tm,Yb,Lu
1380 A4=LGT (A1)-16*N @ MOVE A4,8 @ LABEL "REE ATOMIC NUMBER"
1390 MOVE N1,8 @ DISP "PLOT TITLE".@ INPUT B$@ LORG 4 @ LABEL B$
1400 MOVE LGT (B)/2+LGT (A1)/2,17.5 @ LORG 4 @ LDIR 0 @ IF F1=1 THEN LABEL "SAMP
LE/CHONDRITE" ELSE DISP "YAXIS TITLE" @ INPUT B$@ LABEL B$
1410 CLEAR @ IF Q8=1 THEN GSTORE "OUTLINE"
1420 GOTO 1630 ! ***** FIND MINIMUM AND MAXIMUM INTEGER LIMITS ***************
1430 A=10 @ B=-10 @ CLEAR @ DISP "LOOKING FOR MIN AND MAX VALUES"
1440 FOR J=1 TO 15 @ IF L$(J)="" THEN 1490
1450 FOR I=1 TO RMAX @ IF S(I,J)<0 OR S(I,J)=0 THEN 1480
1460 IF T(I,J)<A THEN A=T(I,J)
1470 IF T(I,J)>B THEN B=T(I,J)
1480 NEXT I
1490 NEXT J @ N=0 @ B=10^B*1.01 @ A=10^A/1.01
1500 B1=B @ IF B<10 THEN 1520
1510 B1=B1/10 @ N=N+1 @ IF B1>10 THEN 1510
1520 B1=CEIL (B1) @ B=B1*10^N @ N=0
1530 A1=A @ IF A>1 AND A<10 THEN 1570
1540 IF A<1 THEN 1560
1550 A1=A1/10 @ N=N+1 @ IF A1>10 THEN 1550 ELSE 1570
1560 A1=A1*10 @ N=N-1 @ IF A1<1 THEN 1560
1570 A1=FLOOR (A1) @ A=A1*10^N @ RETURN
1580 IF A<.0001 THEN 1610 ! DETERMINE AXIS TIC AND LABEL INTERVALS *************
1590 X9=10 @ FOR I=1 TO 8 @ IF ∫∫.0001*X9 THEN S=.00001*X9 @ GOTO 1160
1600 X9=X9*10 @ NEXT I
1610 IF A/.000000001=A\.000000001 THEN S=A @ GOTO 1160
1620 DISP "ENTER EXACT DECADE BOUNDARIES FOR MIN AND MAX VALUES";@ INPUT A,B@ GO
TO 1610
1630 CLEAR @ DISP "ROW TO BE PLOTTED";@ INPUT I ! PLOT DATA ********************
1640 DISP "PEN NUMBER" @ INPUT A@ PEN A
```

```
1650 DISP "DO YOU WANT A PLOT SYMBOL FOR DATA POINTS ?"&I$ @ INPUT A$
1660 CLEAR @ IF Q8=1 THEN DISP "DO YOU WANT A CLEAN PLOT BASE"&I$ @ INPUT B$
1670 IF B$="Y" THEN GLOAD "OUTLINE"
1680 Q7=1 @ Q6=0 @ IF A$="Y" THEN GOSUB 1970 @ GRAPH
1690 FOR J=1 TO 15 @ IF Q6=2 THEN LINE TYPE 4 @ Q6=0
1700 IF S(I,J)=0 THEN 1750
1710 IF S(I,J)<0 THEN LINE TYPE 4 @ T(I,J)=ABS (T(I,J)) @ Q6=2
1720 IF Q7=1 THEN MOVE T(I,J),J
1730 PLOT T(I,J),J @ Q7=0 @ LINE TYPE 1
1740 IF Q6=2 THEN T(I,J)=-T(I,J)
1750 NEXT J @ PEN UP @ IF Q8#1 THEN 1770
1760 ALPHA @ DISP "PRESS 'A/G' KEY TO VIEW PLOT; OR [CONT] TO CONTINUE" @ PAUSE
1770 CLEAR @ DISP " DO YOU WANT TO CHANGE PLOTTER OR REDRAW BASE ?.(Y)" @ DISP
1780 DISP " DO YOU WANT TO FIND A VALUE FOR A REE ON THE CURRENT PLOT ? (V)"
1790 DISP " (this can only be done on external plotter)" @ DISP
1800 DISP " DO YOU WANT A DOT-MATRIX PRINT OF DIAGRAM ? (P)" @ DISP
1810 DISP " DO YOU WANT TO ADD DATA FROM THE KEYBOARD ? (K)" @ DISP
1820 DISP " DO YOU WANT TO EDIT EXISTING DATA ? (E)" @ DISP
1830 DISP " DO YOU WANT TO CHANGE NORMALIZATION FACTOR ? (N)" @ DISP
1840 DISP " DO YOU WANT TO CONTINUE WITHOUT CHANGES ? (C)" @ INPUT A$
1850 IF A$="N" THEN GOSUB 900 @ GOSUB 680 @ A$="Y"
1860 IF A$="Y" THEN RESTORE 1370 @ GOTO 1080
1870 IF A$="K" THEN Q9=1 @ GOSUB 2260 @ GOTO 1630
1880 IF A$="P" THEN 2410
1890 IF A$="V" THEN 2320
1900 IF A$#"E" THEN 1630
1910 Q9=2 @ GOSUB 790
1920 IF S(I,J)<0 THEN T(I,J)=-LGT (ABS (S(I,J)/L(J))) @ GOTO 1940
1930 IF S(I,J)#0 THEN T(I,J)=LGT (S(I,J)/L(J))
1940 CLEAR @ DISP "DO YOU WANT TO EDIT MORE DATA ?"&I$ @ INPUT A$
1950 IF A$="Y" THEN 1910
1960 GOTO 1630 ! ***** SUBROUTINES FOR PLOTTING DATA SYMBOLS ******************
1970 DISP " SYMBOL CAN BE ANY ALPHA-NUMERIC KEY OR 'P' FOR A POLYGON"
1980 DISP " OR 'E' FOR ERROR BARS" @ INPUT A$@ IF A$="E" THEN 2160
1990 IF A$="P" THEN 2050
2000 FOR J=1 TO 15 @ IF L$(J)="" THEN 2040
2010 IF S(I,J)<0 THEN 2040
2020 MOVE T(I,J),J @ LORG 5 @ LDIR 270 @ IF A$="X" OR A$="O" THEN CSIZE 3,1
2030 LABEL A$
2040 NEXT J @ RETURN ! ********* PLOT POLYGON **********************************
2050 DISP "# OF SIDES, OPEN OR SOLID (O OR S), SIZE (mm), AND ROTATION (0 TO 360
)" @ INPUT E1,E$,E3,E4
2060 IF E1<3 THEN 2050 ELSE A1=E4+90 @ A2=360/E1 @ E3=E3/3.63 @ E5=E3 @ A5=A1
2070 FOR J=1 TO 15 @ IF L$(J)="" THEN 2150
2080 IF S(I,J)<0 THEN 2150
2090 MOVE T(I,J),J @ PEN UP @ GRAPH @ DEG @ A1=A5
2100 IF E$="S" THEN E2=.1 ELSE E2=E3
2110 SETGU @ X1=E3*SIN (A1) @ Y1=E3*COS (A1) @ RPLOT X1,Y1,2
2120 FOR E5=E3 TO E2 STEP -.2 @ FOR E=1 TO E1 @ A1=A1+A2
2130 X(E)=E5*SIN (A1) @ Y(E)=E5*COS (A1) @ RPLOT X(E),Y(E),-1 @ NEXT E
2140 NEXT E5 @ ALPHA @ SETUU @ PEN UP
2150 NEXT J @ RETURN
2160 FOR J=1 TO 15 @ IF L$(J)="" THEN 2220 ! ****PLOT ERROR BARS **************
2170 IF S(I,J)<= 0 THEN 2220
2180 DISP "WHAT IS THE 2 SIGMA ERROR (%) FOR ";L$(J) @ INPUT X9@ X9=X9/100
2190 V=10^T(I,J) @ K(J)=LGT ((1-X9)*V) @ X(J)=LGT ((1+X9)*V)
2200 MOVE X(J),J @ LORG 5 @ LDIR 270 @ LABEL "-" @ MOVE X(J),J
2210 PLOT K(J),J,-1 @ LABEL "-"
2220 NEXT J @ RETURN
```

```
2230 FOR J=15 TO 1 STEP -1 @ IF A$(J)="" THEN 2250 ! KEYBOARD DATA INPUT ******
2240 DISP "DO YOU HAVE DATA FOR ";A$(J)&I$;@ INPUT A$@ IF A$#"Y" THEN L$(J)=""
2250 NEXT J @ RMIN=1
2260 I=RMAX+1 @ DISP "SAMPLE NAME ?  ENTER 'END' WHEN THROUGH" @ INPUT A$
2270 IF A$="END" THEN 2310 ELSE RMAX=I
2280 N$(I)=A$ @ FOR J=15 TO 1 STEP -1 @ IF L$(J)="" THEN S(I,J)=0 @ GOTO 2300
2290 DISP "SAMPLE "&N$(I)&", "&L$(J);"=";@ INPUT S(I,J)
2300 NEXT J @ GOTO 2260
2310 F1=0 @ GOSUB 660 @ RETURN
2320 GOSUB 330 ! ***** SUBROUTINE TO DIGITIZE REE VALUES **********************
2330 DISP "FOR WHICH REE DO YOU WANT A VALUE ? ENTER 'END' WHEN FINISHED"
2340 INPUT A$@ IF A$="END" THEN 1770
2350 FOR J=1 TO 15 @ IF L$(J)="" THEN 2370
2360 IF UPC$ (A$)=UPC$ (L$(J)) THEN 2380
2370 NEXT J @ IF J=16 THEN DISP "REE NOT FOUND" @ GOTO 2330
2380 MOVE LGT (A1),J @ DISP "MOVE PEN TO DESIRED VALUE OVER PLOT"
2390 DISP "PRESS 'ENTER' ON PLOTTER TO CONTINUE" @ DIGITIZE X,Y @ X=10^X*L(J)
2400 REE$=L$(J) @ PRINT REE$&" =";X;" FOR SAMPLE "&N$(I) @ CLEAR @ GOTO 2330
2410 GOSUB 330 ! SUBROUTINE TO DUMP PRINTER GRAPHICS ***************************
2420 DISP "INPUT PRINTER TYPE: 0=82905B, -1=82905A, 1=MOST OTHERS" @ INPUT I
2430 DUMP GRAPHICS 0,0,-1,I @ GOTO 1770
2440 END
```